## 6

## Liabilities and the Cash Flow Waterfall

With asset generation complete, this book now turns to the liability side of the model. While the asset side is fairly standard for a level payment amortization, the liability section can vary greatly depending on the unique structure of the transaction. Due to the variability between deals, the liability side needs to be constructed with as much flexibility as possible. This is achieved by breaking down the liabilities into individual components that work similarly and can be moved around quickly.

Before going into the mechanics of how liabilities are paid, it should be understood what constitutes a liability. For a structured transaction, any cost that is to be paid from the cash generated by the assets is a liability. Foremost are costs, which keep the transaction in existence such as trust, servicer, and rating agency fees. Next is interest and principal due to the parties who funded the assets. This section can be very complicated depending on the different risk classes of debt known as tranches. The method in which these tranches are returned principal can also vary between transactions and as will be seen in the next chapter, within a transaction.

## PRIORITY OF PAYMENTS AND THE CASH FLOW WATERFALL

In actual transactions, how the liability structure functions is dictated by the priority of payments section of a term sheet. This section is often referred to as a waterfall because the available cash starts at the top and then trickles down through an ordered series of liabilities. Each section in the written waterfall explains exactly how the cash should be allocated before moving onto the next section of the waterfall.

Conceptually, the movement of cash through the waterfall transfers well to two-dimensional spreadsheet-based modeling because time and order can be organized using rows and columns. The asset side only used rows for time progression, meaning that a starting balance would begin on row 1 and the next period the balance would be adjusted for amortization and displayed on row 2. Although asset interest, prepayments, defaults, and regular amortization do have a logical order, the vertical column arrangement does not necessarily suggest an organization based


FIGURE 6.1 Cash moves through each period from left to right and then down to the next period.
on time or priority. For the liability side of the model and the cash flow waterfall that is created on a spreadsheet, the periods still progress by rows; but the items in columns are arranged by timing. For example, on the Cash Flow sheet starting from the left after asset amortization, where cash is available, the liabilities are organized in priority as one progresses through the columns to the right. Each period, the cash moves from left to right through the waterfall before progressing to the next period one row below. See Figure 6.1 for more detail.

## The Movement of Cash for an Individual Liability

A standard liability has an assumption such as a rate, fixed amount, or a vector of rates on the Inputs or Vectors sheets. This assumption is then integrated into the Cash Flow sheet similar to assets. However, the difference is that each liability will have a certain priority and can either be paid or not paid depending on how much cash is available. In such a system, the concept of "What You Have and What You Need" is essential to understand and make clear.

Barring extremely high default scenarios, every period a certain amount of cash should be available to pay liabilities. The first liability in the priority of payments will have all of the cash that is available from the assets for payment. Once that liability is paid, the cash available is appropriately reduced and the next liability can use the remaining amount for payment. This continues until the waterfall is done and either all of the cash for the period is used or excess remains and is released or trapped.

Each individual liability needs to be set up in a way that shows exactly what is needed, what can be paid based on what is available, what is unpaid, and what (if any) amount remains. While all of these calculations can be done in one cell it is easier for a model operator to see the "movement" of cash by separating each one of these concepts into different cells.

The concept of "What You Have and What You Need" is demonstrated in the Model Builder exercises for this chapter, which are inserted between explanations of the different types of liabilities. Both trends and unique nuances will become evident as one progresses through each type of liability.

## TYPES OF LIABILITIES

A liability is anything that uses cash in the model. While there can be many different types of liabilities the three basic ones that will be examined are fees, interest, and principal.

## Fees

Every structured transaction has some type of fees involved. Servicers charge a fee to service loans. Trusts charge for taking in cash and distributing it correctly, rating agencies charge for performing analyses and assigning ratings. Banks charge for almost anything they do. Also, as will be seen in the later chapters, there are advanced features such as reserve accounts and hedging mechanisms that can incur fees.

Knowing how to calculate the periodic fee can be tricky sometimes because it can vary from deal to deal. Whole dollar amount fees are simple because they are a fixed amount charged per period. However, in a transaction where assets and liabilities taper off over time it is rare to see many whole dollar amount fees. Typically fees can be charged as a percent of the asset or debt balance. Determining the correct basis for a fee is important because a percentage charged against the asset balance versus a debt balance will result in different amounts. As mentioned earlier the debt is normally split up into different tranches, which are a percentage of the assets. A fee charged off of a senior debt balance that is 90 percent of the assets will be lower than a fee charged directly off of the assets.

## MODEL BUILDER 6.1: CALCULATING FEES IN THE WATERFALL

1. Before any specific work can be done on the fees, a liability section needs to be created on the Inputs sheet. On that sheet, merge cells B22:O22 and enter Liability Inputs as the label for that cell. Row 23 will be where the labels are stored for the liability assumptions.
2. Two liability input sections need to be created for this Model Builder exercise. The first section establishes a description for the different types of debt. Project Model Builder uses a senior-subordinated structure, which is explained later in this chapter, but requires two tranches of debt. Enter the assumption label Debt Description in cell B23. Enter the label Senior Debt 1 in cell B24 and name it LiabDes1. Also enter the label Sub Loan in cell B25 and name it LiabDes2. Rows 24 and 25 will contain liability assumptions that correspond to the respective tranche.
3. Project Model Builder has three possible fee inputs. There is one input for each debt tranche and one input that is based on the assets. This demonstrates the difference between calculating a fee against the asset balance or a debt balance. Still on the Inputs sheet, enter the label Fees in cell H23. In cell H24 enter 0.50\% as a value and name the cell LiabFees1. In cell H 25 enter $0.0 \%$ as a value and name the cell LiabFees2. The first liability fee will pertain to the senior debt, while the second fee (which is zero in this case) applies to the subordinate debt.
4. Remain on the Inputs sheet, but jump over to cell B28. This section will be for Structural Assumptions that are part of the deal structure as a whole and are not functions of the debt tranches. Enter the label Structural Inputs in cell B28. Directly below, in cell B29, enter the label Asset Based Fees. To the right, in cell C29, enter the value $2.00 \%$. Name cell C29 AssetFee. At this point the updated part of the Inputs page should look like Figure 6.2.
5. Switch over to the Cash Flow sheet. The last Model Builder exercise left off calculating the total cash available in Column X. Columns Y through AL is skipped for now because they are used later in the next chapter, when the advanced liability structure is introduced. The Fees section starts in column AM. Enter the following labels in their corresponding cells:
AM4: Fees Due
AN4: Fees Paid
AO4: Unpaid
AP4: Cash Remaining
6. The first period calculation for Fees Due begins in cell AM7. There are three distinct fees that need to be paid: the (1) Senior Debt fee, the (2) Subordinate Debt fee, and the (3) Asset Based fee. The Asset Based Fee will be the input rate


FIGURE 6.2 The fee section of the Inputs sheet.
multiplied against the assets' current balance. Remember that the fees typically quoted are annual rates, so the fee needs to be converted to a periodic rate. The Asset Based Fee is calculated by entering the following formula into cell AM7:

$$
=\left(\mathrm{C} 7 * \text { AssetFee }{ }^{*} \mathrm{~L} 7\right)
$$

This formula multiplies the Asset Based Fee assumption by the current asset balance and also by the periodic factor.
7. The other two fees that are due will be added to this formula. The calculation is similar, but there are noticeable differences. Modify the formula in cell AM7 to read as follows:

$$
=(\mathrm{C} 7 * \text { AssetFee*L7 })+(\mathrm{C} 7 * \text { LiabFees1*CB6 })+(\text { C7*LiabFees2*CF6 })
$$

Notice that two more blocks of fees have been added. The first takes the senior debt's fee (LiabFees1) and multiplies it by the senior debt's current balance. The second takes the subordinate debt's fee (LiabFees2) and multiplies it by the subordinate debt's current balance. Both have also been multiplied by the periodic factor (cell C7). Copy and paste the complete formula over the range AM7:AM366.
The value that appears is what is due for the fees, not what is actually paid. Comparing this value to the total cash available for liabilities is the beginning of the concept "What You Have and What You Need." In this case the amounts in column X are "What You Have" and the amounts in column AM are "What You Need."
8. Column AN is where the amount that is actually paid is calculated. A beginning modeler often runs into problems by subtracting what is due from what is available. While this logically makes sense, many problems are created when there isn't enough cash available and the amount paid becomes negative. Then IF statements are introduced and the entire formula becomes messy.

A simple method to use is to always take the least of "What You Have and What You Need." This translates into a MIN formula for the cash available and the amount due. In cell AN7 enter:

$$
=\mathrm{MIN}(\mathrm{X} 7, \mathrm{AM} 7)
$$

Using this technique will ensure that nothing more than what is available can be paid. Copy and paste this formula over the range AN7:AN366.
9. The previous formula only pays from what is available. In stress situations, there could be shortfalls of cash that need to be tracked. Subtracting the amount paid from the amount due will solve that problem. In cell AO7 enter:

$$
=\mathrm{AM} 7-\mathrm{AN} 7
$$

Copy and paste this formula over the range AO7:AO366.
10. Finally, to keep the cash "moving," a column needs to be used to populate the cash remaining. This calculation is the cash that was available prior to the liability minus what was paid from that cash. In cell AP7 enter:

$$
=\mathrm{X} 7-\mathrm{AN} 7
$$

Copy and paste this formula over the range AP7:AP366.
At this point the fee section is complete and should look like the screen in Figure 6.3. Keep in mind the methodology that was developed in this section because any liability that needs to be added to a model can be done so in this fashion. See MB6-1.xls in the Ch06 folder on the CD-ROM for a completed example of these steps.

## Interest

The primary purpose of an entity lending money in a transaction is to generate a return on capital, which is done by charging an interest rate against the money lent. For a private transaction that is not sold into the public markets, the interest rate usually is a bank's funding rate or swap rate plus a margin. If the deal is sold to investors, the rate will be the rate that investors earn on their principal.

Banks that lend money often do this on a floating rate basis using indexes such as LIBOR or prime as the base rate plus a margin. This means that the rates are

|  | AL | AM | AN | AO | AP | A. |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| 1 |  |  |  |  |  |  |
| 2 |  |  |  |  |  |  |
| 3 |  | Fees |  |  |  |  |
| 4 |  | Fees Due | Fees Paid | Unpaid | Cash Remaining |  |
| 5 |  |  |  |  |  |  |
| 6 |  |  |  |  |  |  |
| 7 |  | 166,666.67 | 166,667 | - | 515,286 |  |
| 8 |  | 166,497.65 | 166,498 | - | 531,990 |  |
| 9 |  | 166,300.08 | 166,300 | - | 548,600 |  |
| 10 |  | 166,074.01 | 166,074 | - | 565,107 |  |
| 11 |  | 165,819.48 | 165,819 | - | 581,501 |  |
| 12 |  | 165,536.57 | 165,537 | - | 598,887 |  |
| 13 |  | 165,225.36 | 165,225 | - | 615,031 |  |
| 14 |  | 164,885.94 | 164,886 | - | 631,039 |  |

FIGURE 6.3 The Fee section on the Cash Flow sheet is an introduction to the concept of moving cash and "What You Have vs. What You Need."
susceptible to market fluctuations and can change from period to period. If the assets are generating cash on a fixed rate basis, then there is the possibility of a mismatch between asset and liability rates. For instance, if the assets in a transaction are generating a weighted average rate of 7 percent and LIBOR skyrockets to 10 percent, then a bank that is funding based on LIBOR will lose money. Hedging instruments, for this reason, such as caps and swaps are used. Here the liability side should use the swap rate. Hedging instruments in structured transactions are discussed in more detail later in this book.

The other option to bank funding is investors who typically lend their money by buying bonds that are sold through an investment bank. These bonds are set at a disclosed interest rate that is normally fixed, but can also be floating. With a bond deal a swap is less likely to be needed because the rate can be fixed or set against a floating index.

In both cases, the debt rate is commensurate to the level of risk that the bank or investor is taking. A deal is determined to have a certain level of risk based on the expected loss of the transaction. The rate paid to debt in a deal should be proportionate to the risk of the transaction. However, structured transactions are specifically set up to mitigate and parse risk, so there can be different rates within a transaction.

Managing the risk introduces a concept called credit enhancement, which protects an investor against loss. Credit enhancement can take many forms such as excess interest being generated by the assets, reserve accounts, and/or subordinated debt or equity. Depending on the amount of credit enhancement a transaction will be able to withstand a certain amount of loss. Rating agencies have set certain standards for each asset class to determine a risk rating for a transaction, depending on how the structure holds up to certain stresses. This will be covered later in the text, but for now it is important to understand that the rate debt earns depends on how vulnerable the debt is to loss in the transaction.

## MODEL BUILDER 6.2: CALCULATING INTEREST IN THE WATERFALL

1. For each tranche of debt four inputs need to be known to calculate the correct amount of interest: whether the interest being charged is fixed or floating, the index if it is floating, the fixed rate if it is fixed, and finally the margin on top of either the floating or fixed rate. It may seem confusing to have a margin added to a fixed rate, but sometimes a fixed swap rate is used and a bank will charge a margin. Enter the labels for these four assumptions on the Inputs sheet in the following cells:
D23: Liability Interest Type
E23: Floating Rate Curve
F23: Fixed Rate
G23: Loan Margin
2. The Liability Interest Type for a given debt tranche can be one of three types:

- Floating, when an index is used.
- Fixed, either a fixed bond or swap rate.
- Custom, a rate that changes over time based on the user or a third party. Custom is also used because rating agencies often derive and make available their own stressed interest rate curves.
In the chapter on asset amortization, the range named lstIntType was already created for these three types of interest. Go to the Inputs sheet and create a data validation list in cell D24 using 1stIntType as the range. Name cell D24 LiabIntType1. Repeat this in cell D25, but name that cell LiabIntType2.

3. Cells E24 and E25 need to produce the name of the curve that is being used if the debt is floating rate. This will be done exactly as was done for the interest rates on the asset side. Create a data validation list in cell E24 using lstInterestRates as the range. Name cell E24 LiabLoanIndex1. Repeat this in cell E25, but name that cell LiabLoanIndex2.
4. The fixed rates will be stored in cells F24 and F25 for each tranche. Name cell F24 LiabFxdRate1 and F25 LiabFxdRate2. Keep both of these cells empty for now.
5. The loan margin is the final input that is needed for this section. Name cell G24 LiabMarg1 and G25 LiabMarg2. Enter $\mathbf{1 . 0 0 \%}$ as a value for cell G24 and $0.00 \%$ for cell G25. The Inputs sheet should now look like the screen in Figure 6.4.
6. The next step is to switch over to the Cash Flow sheet and calculate the debt interest. A standard structured transaction will pay the most senior debt interest first. In Project Model Builder columns AR:AX will be used for the Senior Debt Interest. Column AQ will be left blank as a separator.
The first piece of information needed is the correct annual interest rate for the period. Enter the label Note Interest Rate in cell AR4. While step 2 created the possibility of three types of interest rates, there are actually only two options for storing the rates: on the Inputs sheet or on the Vectors sheet. This is important to know because a formula needs to know when and where to look for a specific data point. The simplest situation for rates is when a fixed rate is used, which is stored on the Inputs sheet. To account for that possibility, begin the formula in


FIGURE 6.4 The liability interest rate section complete on the Inputs sheet.
cell AR7 as follows:
= IF(LiabIntType1="Fixed",LiabFxdRate1

This part of the formula checks to see if a fixed rate is selected. If it is, the interest rate should be the fixed rate on the Inputs sheet. However, if the rate is floating or custom it will be stored in a curve on the Vectors sheet. The correct vector to use will then be determined by the floating rate curve that is selected on the Inputs sheet. This requires an OFFSET-MATCH combination as seen before:

$$
\begin{aligned}
= & \operatorname{IF}(\text { LiabIntType } 1=\text { "Fixed",LiabFxdRate1,OFFSET(Vectors!\$D\$6, } \\
& \text { Vectors!A7,MATCH(LiabLoanIndex1,1stInterestRates,0)) }
\end{aligned}
$$

This addition to the formula will select the correct annual rate for the period on the Vectors sheet depending on the name of the curve that is selected in the range LiabLoanIndex1 on the Inputs sheet. Finally adding the margin, if one exists, completes the formula:

$$
\begin{aligned}
= & \text { IF(LiabIntType1="Fixed",LiabFxdRate1,OFFSET(Vectors!\$D\$6, } \\
& \text { Vectors!A7,MATCH(LiabLoanIndex1,lstInterestRates,0))+LiabMarg1) }
\end{aligned}
$$

Copy this formula over the range AR7:AR366.
7. The interest due is easy to calculate once the annual interest rate for the period is known. However, no principal balance information is known until Model Builder 6.3, so an actual value will not show up. A proxy value will be used until that section is complete. Still on the Cash Flow sheet skip over to cell CB6 and enter the value $95,000,000$. This is the assumed starting principal balance of the senior debt for now.

Go back to cell AS4 and enter the label Note Interest Due. In cell AS7 enter the following formula:

$$
=\mathrm{C} 7 * \mathrm{AR} 7 * \mathrm{CB} 6
$$

This formula takes the annual interest rate for the period, converts it to a periodic interest rate and then multiplies that value against the prior period's ending principal balance. It is important to understand the difference between end of period and beginning of period. In this model, the balance referenced is always one row back because that balance is an end of period balance. Always make sure that the balance an interest rate is being applied to is either the end balance for the prior period or the beginning balance of the current period. Copy this formula over the range AS7:AS366. Since there is no principal balance information yet, it is normal that the values for the cells below row 7 will be zero for most of the columns in this section. This will change after Model Builder 6.3.
8. In cell AT4 enter the label Note Interest Paid. This "paying" formula will be similar to the one used in the fee section. Remember the rule, "take the lesser of what is available and what is needed" and enter the following in cell AT7:

$$
=\mathrm{MIN}(\mathrm{AP} 7, \mathrm{AS} 7)
$$

This formula takes the lesser of the cash remaining after fees have been paid and the note interest that is due for the period. Copy this formula over the range AT7:AT366.
9. Column AU will track the unpaid amounts. Enter the label Unpaid in cell AU4. In cell AU7 enter:
= AS7-AT7

As done before in the fee section the paid amount is subtracted from the due amount to determine the unpaid amount. Copy this formula over the range AU7:AU366.
10. Skip over columns AV and AW for now and leave them blank. Those will be reserved for an advanced liability structure in the next chapter. In cell AX4 enter the label Cash Remaining and in cell AX7 enter:

$$
=\mathrm{AP} 7-\mathrm{AT} 7
$$

This formula subtracts the amount paid to interest from the previous cash remaining. Copy this formula over the range AX7:AX366. The senior debt interest section should look like Figure 6.5.
11. The senior debt interest calculations are now complete, but the subordinated debt interest remains unfinished. Subordinated debt is usually lower in the waterfall than many other items, so it will appear further to the right on the Cash Flow sheet. Since Project Model Builder's waterfall is preplanned; the exact columns that the subordinated debt fits into are known. However, keep in mind that when creating a scratch model the final columns, where liabilities end up, may not be clear as it is being built and may require inserting or deleting columns.

To complete the subordinated debt, stay on the Cash Flow sheet and move over to column BN . All of the formulas will be very similar to the senior debt interest formulas, so this section should be quick to make. Enter the following labels in their respective cells:
BN4: Sub Loan Rate
BO4: Loan Interest Due
BP4: Loan Interest Paid
BQ4: Unpaid Interest
BR4: Cash Remaining


FIGURE 6.5 The nearly complete senior interest section of the Cash Flow sheet.
12. As with the senior debt balance, a proxy value should be entered for the subordinate debt. Enter 5,000,000 in CF6. Next, enter the following formulas in the cells as noted:

```
BN7: = IF(LiabIntType2 = "Fixed",LiabFxdRate2,OFFSET(Vectors!$D$6,
    Vectors!A7,MATCH(LiabLoanIndex2,IstInterestRates,0))
    +LiabMarg2)
BO7: = C7*BN7*CF6
BP7: = MIN(BL7,BO7)
BQ7: = BO7 - BP7
BR7: = BE7 - BP7
```

Copy the range BN7:BR7 and paste it over the range BN7:BR366. Do not be concerned if many of the cells have zero values. The cash flow waterfall is being constructed using a conceptual methodology, not ordinal. This requires many blank and zero value cells until the entire waterfall is complete.
Also, a final note on interest relates to unpaid amounts. The example model does not capitalize unpaid interest nor does it make the unpaid interest due the next period. Many transactions are structured this way and the modeling should reflect such details. Also, as with the other Model Builder sections, the Ch06 folder on the CD-ROM features a corresponding completed example.

## Principal

In addition to interest, banks and investors expect to have the principal amount they loaned returned. Principal is often returned at different priority levels to mitigate and parse risk. Earlier it was briefly mentioned that there are different risk rated classes of debt-tranches. The way to think about the debt structure is that assets must always equal liabilities. Assets are not free and must be 100 percent funded from the start; however, an investor may not want to take 100 percent of the risk that the assets do not pay back all of the investors' loaned principal. Instead, a bank could sell bonds equal to 90 percent of the assets as senior debt and the other 10 percent as subordinated debt. The reason the first 90 percent is considered to be senior is because it has priority to its principal over the subordinated debt in the cash flow waterfall. Such a set up is also known as a senior subordinated structure.

Senior debt should always have priority when receiving principal versus subordinated debt; but there are two different methods of amortization that differ in regards to tranche principal repayment: sequential and pro rata. A sequential pay method pays the entire senior principal balance before paying one dollar of the subordinated debt. This means that it could be months or years in a deal until the subordinated debt receives a principal payment. It also makes the senior debt more secure because the subordinate debt does not decrease and, as discussed later, is a source of credit enhancement for the senior debt.

The other type of principal payment methodology is pro rata, which as the name implies pays principal proportionately. A simple example is if there is $\$ 100$ of assets, funded by a senior loan of $\$ 90$ and a subordinate loan of $\$ 10$. The proportion of the debt is 90 percent senior loan and 10 percent subordinate loan. If $\$ 5$ of principal came in during a period then the senior loan would be due $\$ 4.50$ ( $\$ 5 \% 90$ percent) and the subordinate loan due $\$ .50$ ( $\$ 5^{*} 10$ percent). While fixed during normal performance, these proportions can change within a deal if the assets are incurring unexpectedly high levels of default.

However, a change to principal allocation within a deal is a more advanced concept that is discussed in further detail later in the book. At this point, the focus is to understand the basic flow of principal through the cash flow waterfall. Project Model Builder uses a senior-subordinated debt structure with the option for either sequential or pro rata principal payment. The model is also set up with the concept that principal is "passed through" to the debt. This means that if the transaction is performing as expected any amortization of the assets should directly result in the same amortization of the debt.

## MODEL BUILDER 6.3: CALCULATING PRINCIPAL IN THE WATERFALL

1. Go to the Inputs sheet and enter the label Advance Rate in cell C23. The advance rate is the debt principal amount expressed as a percentage of the assets. If there is $\$ 100$ of debt and the senior debt is $\$ 95$ at day one, then the advance rate for
the senior debt is 95 percent. In cell C24 enter 95.00 percent. Name cell C24 LiabAdvRate1. Since there is only one other tranche of debt the subordinate amount advanced will always be 100 percent minus the senior advance rate. Enter the following formula in cell C25:

$$
=1-\text { LiabAdvRate } 1
$$

Name cell C25 LiabAdvRate2.
2. The other necessary input is the principal payment or allocation type. There are only two types discussed so a data validation list works well. Go to the Hidden sheet and enter the label PrinType in cell A21. Enter Sequential in cell A22 and Pro rata in A23. Name the range A22:A23 1stPrinType. Go back to the Inputs sheet and enter the label Prin Allocation Type in cell J23. Create data validation lists in cells J24 and J25 using lstPrinType as the range. Name cell J24 LiabPrinType1 and J25 LiabPrinType2. So far the section should look like Figure 6.6.
3. Now is the time to change the proxy values for the principal balances that were created in Model Builder 6.2. Go to the Cash Flow sheet and label the following cells:
CB4: Senior Loan EOP Balance
CC4: Senior Interest
CD4: Senior Principal
The initial senior principal balance will be the advance rate multiplied by the initial asset balance. After the first period the balance will be reduced commensurate to the assets. Since there are two possible states, initial period and after, an IF statement formula is needed. In cell CB6 enter:

$$
=\mathrm{IF}(\mathrm{~A} 6=0, \mathrm{~V} 6 * \text { LiabAdvRate1,CB5-CD6) }
$$

This formula checks to see if the period is the initial period, multiplies the advance rate by the asset balance if it is the initial period, or subtracts the current principal payment from the prior period's balance if the period is anything else than 0 . Copy this formula over the range CB6:CB366.


FIGURE 6.6 The principal section of the liabilities on the Inputs sheet.
4. The same should be done for the Sub Loan. Enter these labels in the following cells:
CF4: Sub Loan EOP Balance
CG4: Sub Interest
CH4: Sub Principal
The only difference in the formula are the references and that LiabAdvRate2 is used instead of LiabAdvRate1. Cell CF6 should have the following formula:

$$
=\mathrm{IF}(\mathrm{~A} 6=0, \mathrm{~V} 6 * \text { LiabAdvRate } 2, \mathrm{CF} 5-\mathrm{CH} 6)
$$

Copy this formula over the range CF6:CF366. Leave rows 7:366 for columns CC, CD, CG, and CH empty for now.
5. Next the senior debt principal amounts will be calculated in their correct place in the waterfall. Typically senior principal pays after senior interest. Still on the Cash Flow sheet, enter these labels in the following cells:
AZ4: Principal Due
BA4: Principal Paid
BB4: Unpaid
BE4: Cash Remaining
Leave columns BC and BD blank for now.
6. For now the senior tranche principal due can be either sequential, where all of the asset amortization is due to the senior tranche first or pro rata, where the senior tranche's proportional share of the asset amortization is due. Enter the following formula in cell AZ7:

$$
=\mathrm{IF}(\text { LiabPrinType } 1=\text { "Sequential", } \mathrm{MIN}((\mathrm{~N} 7+\mathrm{Q} 7+\mathrm{R} 7), \mathrm{CB} 6)
$$

Quite a bit is taking place in this formula. First notice that an IF statement is used to check what type of principal allocation method is being used for the senior debt. If a sequential method is used, then the principal due to the senior notes will be the amount that the assets amortized in that period. That amount consists of scheduled amortization, voluntary prepayments, and new defaults (columns $\mathrm{R}+\mathrm{Q}+\mathrm{N}$ respectively).

A point of confusion that comes to even a seasoned structured finance professional is why new defaults are included in the debt principal due calculation. This is the heart of one of the forms of risk mitigation: using excess cash or spread to cover loss. Since the assets have been reduced by the defaults, the debt will need to be reduced by the same amount. The debt of course has access to prepayments and scheduled amortization, which provide cash to the waterfall, but defaults are noncash generating amounts. If there were no other cash in the deal besides the prepayments and scheduled amortization, then the debt principal due could never be paid because the defaulted amount would make the debt principal due calculation too high.

However, most transactions are structured so the assets generate more interest than is due to fees and debt interest. This concept is officially known as excess
spread. That extra money will trickle down along the waterfall and eventually be available to pay principal due. Since the defaulted amount is built into the debt principal due calculation, that extra amount can be paid if there is excess spread in the transaction. This is how excess spread is typically used to first cover losses. If there is no excess spread then other sources of credit enhancement are necessary to cover the defaulted amount, which will be seen in the next chapter. Going back to the formula, also notice that there is a MIN function for the asset amortization amounts and the debt's prior period ending balance. The MIN function prevents the principal due from exceeding the balance of the debt. This typically occurs in the final period when the debt balance is small and possibly lower than the amount the assets amortized.
7. Complete the formula in AZ7 by adding the following shown in bold:

$$
\begin{aligned}
= & \operatorname{IF}(\text { LiabPrinType } 1=\text { "Sequential" }, \mathrm{MIN}((\mathrm{~N} 7+\mathrm{Q} 7+\mathrm{R} 7), \mathrm{CB} 6), \\
& \operatorname{MIN}((\mathrm{N} 7+\mathrm{Q} 7+\mathrm{R} 7) * \text { LiabAdvRate1,CB6)) }
\end{aligned}
$$

This addition to the formula is for a pro rata principal allocation method. Instead of using the entire asset amortization amount for the period, the formula takes a percentage of the asset amortization amount. The MIN function is also used here to cap the principal due by the prior period's debt balance. Copy and paste the complete formula over the range AZ7:AZ366.
8. The remaining calculations revert back to the concept of "What You Have and What You Need." For BA7 enter the familiar MIN formula:
= MIN(AX7,AZ7)

This will take the lesser of the cash remaining after interest was paid and the amount due for principal. Copy and paste the formula over BA7:BA366.
9. In cell BB7 enter:
= AZ7-BA7

This subtracts the principal paid from the principal due and displays any unpaid amounts. Copy and paste the formula over range BB7:BB366.
10. Enter the following formula in BE7 to determine the cash remaining:

$$
=\mathrm{AX} 7-\mathrm{BA} 7
$$

Copy and paste the formula over BE7:BE366. The screen should now look like Figure 6.7.
11. Still on the Cash Flow sheet, move across to column BT. Add the following labels
BT4: Loan Principal Due
BU4: Loan Principal Paid

## BV4: Unpaid

BW4: Cash Remaining
A similar debt principal calculation needs to be done for the subordinated debt; but there is a major difference. In a sequential principal allocation system, the subordinated tranche should not receive any principal until the senior tranche is completely paid off. Modeling this logic is achieved through an IF-AND combination in cell BT7:

$$
=\operatorname{IF}(\text { AND }(\text { LiabPrinType } 2=" \text { Sequential" }, \mathrm{CB} 6>0), 0,
$$

The beginning of the formula for BT7 will return a 0 value if the principal allocation system is set to "Sequential" and the senior debt has a principal balance. If the senior debt is paid off, then another IF statement is required because a FALSE value for the first IF statement can be caused by either a different principal allocation type or a paid off senior tranche. Add the following shown in bold to cell BT7:

$$
\begin{aligned}
= & \text { IF(AND(LiabPrinType } 2=\text { "Sequential", CB6>0), } 0, \\
& \operatorname{IF}(\text { LiabPrinType } 2=\text { "Sequential" }, \mathrm{MIN}((\mathrm{~N} 7+\mathrm{Q} 7+\mathrm{R} 7), \mathrm{CF} 6),
\end{aligned}
$$

| A | AZ | BA. | BB | BC | $B D$ | BE | B |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| 1 |  |  |  |  |  |  |  |
| 2 |  |  |  |  |  |  |  |
| 3 | Senior Debt P | rincipal |  |  |  |  |  |
| 4 | Principal Due | Principal Paid | Unpaid |  |  | Cash Remaining |  |
| 5 |  |  |  |  |  |  |  |
| 6 |  |  |  |  |  |  |  |
| 7 | 96,342 | 96,342 | - |  |  | 94,360 |  |
| 8 | 112,613 | 112,613 | - |  |  | 78,018 |  |
| 9 | 128,862 | 128,862 | - |  |  | 73,432 |  |
| 10 | 145,080 | 145,080 | - |  |  | 68,772 |  |
| 11 | 161,259 | 161,259 | - |  |  | 64,040 |  |
| 12 | 177,392 | 177,392 | - |  |  | 60,345 |  |
| 13 | 193,468 | 193,468 | - |  |  | 55,465 |  |
| 14 | 209,481 | 209,481 | - |  |  | 50,511 |  |
| 15 | 225,422 | 225,422 | - |  |  | 45,483 |  |
| 16 | 241,282 | 241,282 | - |  |  | 40,381 |  |
| 17 | 257,054 | 257,054 | - |  |  | 35,205 |  |
| 18 | 272,728 | 272,728 | - |  |  | 29,955 |  |
| 19 | 288,298 | 288,298 | - |  |  | 19,682 |  |
| 20 | 303.754 | 303,754 | - |  |  | 9.336 |  |

FIGURE 6.7 The Senior Principal section of the Cash Flow sheet.

Now if the principal allocation is set to "Sequential" and the senior tranche is paid off then the subordinate tranche will continue to receive $100 \%$ of the asset amortization. Finally, the last part (shown in bold) that needs to be added is when the principal allocation method is set to "Pro rata":

$$
\begin{aligned}
= & \mathrm{IF}(\text { AND }(\text { LiabPrinType } 2=\text { "Sequential",CB6>0), } \\
& 0, \mathrm{IF}(\text { LiabPrinType2 }=\text { "Sequential", MIN((N7+Q7+R7),CF6), } \\
& \operatorname{MIN}((\mathrm{N} 7+\mathrm{Q} 7+\mathrm{R} 7) * \text { LiabAdvRate2,CF6))) }
\end{aligned}
$$

Similar to the senior tranche, when the principal allocation method is pro rata then the subordinate tranche will only pay down by its proportionate share of the asset amortization. Copy and paste the complete formula over the range BT7:BT366.
12. The rest of the calculations should seem familiar by now-so in the following cells enter:
BU7: = MIN(BR7,BT7)
BV7: = BT7-BU7
BW7: = BR7-BU7
Copy the range BU7:BW7 and paste it over the range BU7:BW366.
13. With all of the interest and principal calculations in place, the debt balances can be completed. At this point many of the columns that have zero values will change to real values since the debt balances will extend over time.
In cell CC7, reference the senior interest that has been paid for the period by entering $=$ AT7. In cell CD7, reference the senior principal that has been paid for the period by entering = BA7. Copy and paste CC7:CD7 over the range CC7:CD366. The same should be done for the subordinated debt. In cell CG7 enter $=$ BP7 and in CH7 enter $=$ BU7. Copy and paste cells CG7:CH7 over the range CG7:CH366.

This completes the debt principal calculations. At this point, the debt principal should be decreasing as principal payments come in. In fact, the basic liability waterfall is complete. However, the waterfall is not operational because a few advanced structures are missing. Also keep in mind that this is one of many unique liability structures. To accurately model a transaction, the priority of payments needs to be thoroughly understood. Refer to MB6-3.xls in the Ch06 folder on the CD-ROM for a complete example of this section.

## UNDERSTANDING BASIC ASSET AND LIABILITY INTERACTIONS

With the creation of the basic liability structure, the value of modeling a transaction begins to become clear. Assumptions can be made that replicate the structure and behavior of assets, which generate cash. The amount and timing of the cash depends on the assumptions for asset amortization, prepayments, defaults, and recoveries.

There are endless possibilities for the amount and timing of the available cash in the transactions. The same variability can be seen on the liability side in the cash flow waterfall.

Different liability structures can be put in place to work with any given pool of assets. Stress scenarios are then run to see how the liability structure handles defaulted assets. Up to this point, the only form of protection against loss or credit enhancement is excess spread. When excess spread is not enough to help pay the liabilities off by the final maturity of the transaction, then the debt holders will sustain a loss. To a lesser extent, there can be situations where interest is not completely paid and the debt holders receive less return than anticipated.

While excess spread in a transaction is an excellent source of protection, structured transactions have developed multiple methods of protecting against stressed situations. These methods add another level of complexity to a model, but must be incorporated to accurately model a transaction. The next chapter explains these advanced cash flow structures and shows how to incorporate them into a model.

